
PART  ONE

Introduction

CHR  = Constraint Handling Rules

� CHR is a very high level  programming language
� based on rules

� propagation rules:
� clouds Þ  forecast(rainy).
� forecast(rainy) Þ  bring(coat).
� forecast(sunny) Þ  bring(sunscreen).

� simplification rules:
� bring(coat), bring(sunscreen) Û  bring(umbrella).

� stand-alone (CHR-only)  or extending a host language

More information:
Peter Van Weert's talk on Thursday

Syntax of CHR

� Propagation rule:

head   Þ   guard   |   body .

example:   dist(A,D), road(A,B,L)   Þ  dist(B,D+L).

� Simplification rule:

head   Û   guard   |   body .

example:    dist(A,X), dist(A,Y)   Û   X�Y |  dist(A,X) .

head : CHR constraints
guard : host language (built-in)

body : CHR constraints + host language

Operational semantics of CHR

� Propagation rule:  … ADD body  TO STORE

head   Þ   guard   |   body .

example:   dist(A,D), road(A,B,L)   Þ  dist(B,D+L).

� Simplification rule:  … REPLACE head  BY body

head   Û   guard   |   body .

example:   dist(A,X), dist(A,Y)   Û   X�Y |  dist(A,X) .

IF head  IN STORE (AND guard  HOLDS), THEN...

Features of CHR

� Propagation rule:

head   Þ   guard   |   body .

example:    dist(A,D), road(A,B,L)   Þ  dist(B,D+L).

� Simplification rule:

head   Û   guard   |   body .

example:   dist(A,X), dist(A,Y)   Û   X�Y |  dist(A,X) .

simplification = equivalence

propagation = implication

Logical semantics
CHR has a declarative semantics!

PART  TWO

Writing CHR programs

CHR(Prolog) by example

� Simple example: color mixing in CHR
� We first declare CHR constraints as follows:

:- chr_constraint red, blue, yellow, purple, …

� Then we write the rules:
red, blue <=> purple.

blue, yellow <=> green.

yellow, red <=> orange.

CHR(Prolog) by example

� Simple example: color mixing in CHR
red, blue <=> purple.

blue, yellow <=> green.

yellow, red <=> orange.

� CHR program execution:

� user gives a goal
� rules are applied exhaustively
� the remaining constraints are the result

CHR(Prolog) by example

� Simple example: color mixing in CHR
red, blue <=> purple.

blue, yellow <=> green.

yellow, red <=> orange.

� Example interaction:
?- blue, red.

purple

?- yellow, blue, red.

green

red

Refined semantics

Execution from left to right and from 
top to bottom (cf. Prolog)

Why this 
answer?
(and not, say, 

“yellow, purple”)

  

36

Constraints with arguments

� Add anything to brown and it remains brown:
red, blue <=> purple.

blue, yellow <=> green.

yellow, red <=> orange.

brown, red <=> brown.

brown, blue <=> brown.

brown, yellow <=> brown.

brown, purple <=> brown.

...

Constraints with arguments

� This becomes a bit tedious, can't we write 
something like this instead?
brown, _ <=> brown.

� The above will not work in CHR (but it does work in a 
related formalism called ACD term rewriting)

� But we can write our program in a different way...

  

38

Constraints with arguments

� From many 0-ary constraints to one unary constraint :
:- chr_constraint red, blue, yellow, purple, ...

red, blue <=> purple.

blue, yellow <=> green.

yellow, red <=> orange.

:- chr_constraint color/1.

color(red), color(blue) <=> color(purple).

...

Constraints with arguments

� Now we can write more general rules:
:- chr_constraint color/1.

color(X), color(Y) <=> mix(X,Y,Z) | color(Z).

color(brown), color(_) <=> color(brown).

% host language

mix(red,blue,purple).

mix(blue,yellow,green).

mix(yellow,red,orange).

Type and mode declarations

� Optionally, we can specify types and modes:
% no type/mode declaration:

:- chr_constraint color/1 .

% only mode declaration:

:- chr_constraint color(+) . % ground argument

% type and mode declaration:

:- chr_constraint color(+colorname) .

:- chr_type colorname ---> red ; blue ; yellow ;...

Simpagation rules

� So far we have only used simplification rules.
� Simpagation rules can be more concise/efficient:

% simplification rule:

color(brown), color(_) <=> color(brown).

% simpagation rule:

color(brown) \ color(_) <=> true.

“ true ” ?

In Prolog, “ true ” is a built-in that does not do 
anything. We use it to indicate an empty body.

Typical pattern #1: flattening lists

� We want to convert “ colors([red,green,blue]) ” 
to “ color(red), color(green), color(blue) ”
:- chr_constraint color(+colorname).

:- chr_type colorname ---> red ; blue ; yellow ;...

:- chr_constraint colors(+list(colorname)).

:- chr_type list(T) ---> [] ; [T|list(T)].

colors([]) <=> true.

colors([C|Rest]) <=> color(C), colors(Rest).

(just like how you would do this in Prolog)

More complex color mixing

� Now we also specify the amount of paint:
:- chr_constraint color(+colorname,+amount).

:- chr_type colorname ---> red ; blue ; yellow ;...

:- chr_type amount == float.     % in liters

(“ float ” is a built-in type for floating point numbers)

Typical pattern #2: “default constructor”

� For backwards compatibility, we still have color/1
:- chr_constraint color(+colorname).

:- chr_constraint color(+colorname,+amount).

:- chr_type colorname ---> red ; blue ; yellow ;...

:- chr_type amount == float.

% we assume 1 liter of paint:

color(C) <=> color(C,1).

Typical pattern #3: maintaining a sum

:- chr_constraint color(+colorname,+amount).

color(C,A1), color(C,A2) 

<=> TA is A1+A2, color(C,TA).

color(C,0) <=> true.

color(X,A1), color(Y,A2) 

<=> mix(X,Y,Z) | TA is A1+A2, color(Z,TA).

CHR(Prolog) one-liners (1)

� Finding the minimum:

min(A ) \ min(B) <=> A =< B | true.

?-  min(8), min(3), min(6), min(7).
min(3)

� Computing the sum:
sum(A), sum(B) <=> C is A+B, sum(C).

?-  sum(3), sum(5), sum(6).
sum(14)

CHR(Prolog) one-liners (2)

� Transitive closure

:- op(700,xfx,before).
:- chr_constraint before(+any,+any).

A before B, B before C ==> A before C.

?- a before b, b before c, c before d.
a before b
b before c
c before d
a before c
a before d
b before d

CHR(Prolog) one-liners (3)

� Naive merge-sort in O(n 2) time

:- op(700,xfx,before).
:- chr_constraint before(+any,+any).

A before B \ A before C <=> B @< C | B before C.

?- 0 before foo, 0 before bar, 0 before baz, 0 before quux.
0 before bar
bar before baz
baz before foo
foo before quux

CHR(Prolog) two-liners (1)

� Greatest common divisor 
(Euclid's algorithm)

:- chr_constraint gcd(+int).

gcd(0) <=> true.

gcd(N) \ gcd(M) <=> N =< M | L is M mod N, gcd(L).

?- gcd(94017), gcd(1155), gcd(2035).
gcd(11)

CHR(Prolog) two-liners (2)

� Prime number generator 
(sieve of Eratosthenes)

:- chr_constraint prime(+int).

prime(N) ==> N>2 | M is N-1, prime(M).

prime(A) \ prime(B) <=> B mod A =:= 0 | true.

?- prime(10).
prime(2)
prime(3)
prime(5)
prime(7)

CHR(Prolog) two-liners (3)

� Fibonacci numbers

:- chr_constraint fib(+int,+int), upto(+int).

upto(_) ==> fib(0,1), fib(1,1).

upto(Max), fib(N1,M1), fib(N2,M2)
  ==> Max>N2, N2 is N1+1 | 
    N is N2+1, M is M1+M2, fib(N,M).

?- upto(10).
fib(10,89)
fib(9,55)
fib(8,34)
fib(7,21)
fib(6,13)
...

CHR(Prolog) two-liners (5)

� Soduko puzzle solver in CHR
:- chr_constraint given(+pos,+val), maybe(+pos,+lis t(val)).

given(P1,V) \ maybe(P2,L)
  <=> sees(P1,P2), select(V,L,L2) | maybe(P2,L2).

maybe(P,L) <=> member(V,L), given(P,V).

sees(X-_, X-_). % same row
sees(_-X, _-X). % same column
sees(X-Y, A-B) :- X//3 =:= A//3,  Y//3 =:= B//3. % s ame box

?- given(1-1,5),given(1-2,3), ..., maybe(1-3,[1,2,3,.. .,9]), ...
given(a-1,5)
given(a-2,3)
given(a-3,7)
...

One last example...

� Simple less-than-or-equal constraint solver
:- op(700,xfx,leq).

:- chr_constraint leq/2.

reflexivity @  X leq X <=> true.

idempotence @  X leq Y \ X leq Y <=> true.

antisymmetry @  X leq Y, Y leq X <=> X=Y.

transitivity @  X leq Y, Y leq Z ==> X leq Z.

?- A leq B, B leq C, C leq A.
A = B
B = C

�

Differences between CHR and Prolog

CHR
basic elements predicates constraints

elements are defined by clauses rules
syntax head :- body. head <=> guard | body.
#heads 1 1, 2, 3, …

unification matching + guard

committed-choice

no applicable definition failure suspension (delay)

Prolog

definition selection 
condition

different applicable 
definitions

try alternatives 
(backtracking)

�  partial result

Committed-choice – different from Prolog!

� In Prolog, backtracking  (proof search) is used to find 
a non-failing derivation

� In CHR there is no backtracking
� :- chr_constraint chr/0, output/1.

chr <=> output(foo).
chr <=> output(bar).
prolog :- output(foo).
prolog :- output(bar).

?- prolog. ?- chr.
output(foo) ; output(foo)

output(bar)

Head matching – different from Prolog!

� In Prolog, unification  is used to match clause heads

� In CHR, matching (one-way unification) is used
� :- chr_constraint chr/1, output/1.

chr(foo) <=> output(bar).
prolog(foo) :- output(bar).

?- prolog(foo). ?- chr(foo).
output(bar) output(bar)

?- prolog(Variable). ?- chr(Variable).
output(bar) chr(Variable)
Variable = foo

?- prolog(quux). ?- chr(quux).
No chr(quux)

PART  THREE

Theory & Applications

History of CHR: some milestones

1991 CHR is born, Thom Frühwirth

1995 Christian Holzbaur implements  CHR(SICStus)
1998 confluence, program analysis (PhD Slim Abdennadher)

2002 Tom Schrijvers implements  Leuven CHR system
2002- optimized compilation (PhDs Gregory Duck, Tom Schrijvers)

2003 First CHR book [ Frühwirth&Abdennadher, Essentials of Constraint Programming]

2004 refined semantics, Gregory Duck et al.

2004 First CHR workshop
2005-  computational complexity (PhD Jon Sneyers)

2005- Peter Van Weert implements  Leuven JCHR (Java)
2007 Sulzmann & Lam implement  first concurrent system

2009 Second CHR book, sixth CHR workshop

Theory topics (1)

� Semantics
� Declarative (logical) semantics

� Classical logic (Frühwirth)
� Linear logic (Hariolf Betz)

� Transaction logic, …
� Compositional semantics (Gabbrielli et al)

� Operational semantics
� Abstract semantics
� Refined semantics (Duck et al)

� Priority semantics (Leslie De Koninck)

Theory topics (2)

� Extensions / variants of CHR
� Adaptive CHR (Armin Wolf)

� Disjunction, search (Abdennadher, Wolf, De Koninck, …)

� Negation, aggregates (Van Weert & Sneyers, …)

� Modularity, solver hierarchies (Duck et al, Schrijvers et al, 
Fages et al) 

� Probabilistic CHR (Frühwirth et al, Sneyers et al)

� ...

Theory topics (3)

� Relationship to other formalisms
� Term rewriting (ACD term rewriting, Duck, Stuckey et al)

� Production rules / business rules (Van Weert)

� Join-Calculus (Sulzmann and Lam)

� Logical Algorithms (De Koninck)

� Graph Transformation Systems (Raiser)

� Petri nets (Betz)

� ...

Computational Complexity Theory

� How does an algorithm scale with the input size?

algorithm A algorithm B

log-linear quadratic

input size

Leuven 5000 2 ms 25 ms

Brussels 50000 23 ms 2.5 seconds

New York City 277863 151 ms 1 min 17 seconds

Florida 1228116 747 ms  25 min, 8 seconds

North America 29883886 22 seconds 10 days, 8 hours, 4 min

O(n log n) O(n2)

Further reading...

� Book:  Thom Frühwirth,  Constraint Handling Rules ,  
Cambridge University Press, July 2009.

� Introductory survey:  Thom Frühwirth,  Theory and 
Practice of Constraint Handling Rules , Special Issue 
on Constraint Logic Programming (P. Stuckey and K. Marriott, 
Eds.), Journal of Logic Programming, Vol 37(1-3), O ctober 1998. 

� Advanced survey:  Jon Sneyers, Peter Van Weert, Tom 

Schrijvers and Leslie De Koninck,  As Time Goes By: 
Constraint Handling Rules — A Survey of CHR 
Research from 1998 to 2007 , Theory and Practice of 
Logic Programming, 2009, To appear.


